
INTRODUCTION
There are various ways on how to implement the
Analog-to-Digital (A/D) conversion in a circuit. For a
simple and low bandwidth analog application such as
DC voltmeter, it is desirable to have a low cost yet high
resolution A/D converter. 

This application note describes a method to implement
A/D conversion on the PIC16C5X and PIC16F5X
series of microcontrollers. The converter requires only
five external components and is software and hardware
configurable for conversion resolutions from six bits up
to ten bits, and conversion times of 250 us or longer.
The method is usable for both voltage and current con-
version and uses a software calibration technique that
compensates for time and temperature drift, as well as
component errors. Following are reasons why
PIC16C5X and PIC16F5X microcontroller families are
ideal for simple analog applications:

• Very low cost
• Few external components required
• Fully programmable; PIC16C5X microcontrollers 

for One-Time-Programmable (OTP) EEPROM 
devices and PIC16F5X for Flash devices

• Available off-the-shelf from distributors
• Calibration in software for improved measurement 

accuracy
• Power savings using Sleep mode
• Output pins have large, current source/sink 

capability to drive LEDs directly

THEORY OF OPERATION
Figure 1 shows a simplified schematic of the A/D
converter circuit. There are two input voltages
connected one at a time to op amp U1. VREF is the fixed
reference voltage used in calibration and VMEAS is the
unknown voltage to be converted. Resistor R1 and
Capacitor C1 form a charging circuit used to convert
input voltage to time. The existence of U1 in the circuit
removes the logarithmic characteristic that would occur
if the input voltage is directly applied to R1 and C1. The
microcontroller controls the U1 operation by turning
ON/OFF the four switches, S1-S4 of Analog Switch IC,
at different conversion stages. Additionally, the
microcontroller measures the time and calculates the
digital representation of the unknown input voltage.

The circuit can also be used as a Current mode A/D
converter. In this case, the input voltage to the current
converter is not needed and the reference current and
input current are both routed via analog switches
directly into the capacitor.

FIGURE 1: A/D CONVERTER CIRCUIT

In order to visualize the different stages of conversion,
let us refer to U1 output voltage VO waveform shown in
Figure 2.

FIGURE 2: OPERATIONAL AMPLIFIER OUTPUT VOLTAGE WAVEFORM

At t0-t1, S1 and S3 are ON, S2 and S4 are OFF and
RA0 is pulled to ground by the software. This yields the
equivalent circuit in Figure 3. VO is equal to VREF since
VIN is equal to VREF and S3 force unity gain feedback.
C1 is discharging or is initially discharged after the
Reset state. In any case, this stage ensures that C1 is
fully discharged before going to the next stage. At the
end of t1, S1 remains ON, S2 remains OFF, S3 is OFF,
S4 is ON and RA0 is configured as an input pin. This
yields the equivalent circuit in Figure 4. As a function of
VREF, VO is started to ramp-up linearly while C1 is
charging. The VO ramp-up continues until the threshold
voltage input Vth of the microcontroller trips. This
generates a software calibration value equal to tref.
This calibration value is measured and used to
calibrate out most circuit errors, including inaccuracies
in the resistor and capacitor, changes in the Vth, and
temperature variation.

FIGURE 3: EQUIVALENT CIRCUIT 
DURING DISCHARGING

FIGURE 4: EQUIVALENT CIRCUIT 
DURING MEASUREMENT

After the software calibration value is measured at t2,
S2 and S3 are ON, S1 and S4 are OFF and RA0 is
pulled to ground again by the software. This yields the
same equivalent circuit in Figure 3. However, VO is
equal to VMEAS since VIN is equal to VMEAS and S3
force unity feedback. C1 is discharging from t2 to t3. At
the end of t3, S2 remains ON, S1 remains OFF, S3 is
OFF, S4 is ON and RA0 is configured as an input pin.
This yields the same equivalent circuit in Figure 4. As a
function of VMEAS, VO is started to ramp-up linearly
while C1 is charging. The VO ramp-up continues until
the Vth of the microcontroller trips. This generates a
software VMEAS value equal to tmeas. This value is
compared to the software calibration value to
determine the actual digital representation of VMEAS.

A/D CONVERTER EQUATIONS
Based on the circuit operation, equation is used by the
microcontroller in order to calculate the final conversion
result.

In Figure 4, current through R1 is equal to the current
through C1.

When input voltage VIN is equal to VREF, the relation
between the two currents are represented as (see
Equation 1):

EQUATION 1: R1 AND C1 CURRENT 
EQUATION WHEN INPUT 
VOLTAGE IS VREF

and when VIN is equal to VMEAS, the relation between
the two currents are represented as (see Equation 2):

EQUATION 2: R1 AND C1 CURRENT 
EQUATION WHEN INPUT 
VOLTAGE IS VMEAS

Integrating Equation 1 and 2 results (see Equation 3
and Equation 4):

EQUATION 3: INTEGRATION RESULT OF 
EQUATION 1

and

EQUATION 4: INTEGRATION RESULT OF 
EQUATION 2

Since VREF and VMEAS is constant input, Equation 3
and Equation 4 can be reduced further to:

EQUATION 5: U1 VOLTAGE OUTPUT 
EQUATION AS A FUNCTION 
OF VREF 

and

EQUATION 6: U1 VOLTAGE OUTPUT 
EQUATION AS A FUNCTION 
OF VMEAS

At the end of each measurement, VO of Equation 5 and
Equation 6 are both equal to Vth, therefore, equating
both equations yields:

EQUATION 7: RESULT OF EQUATIONS 5 
AND 6

In Equation 7, R1 and C1 can be eliminated and solved
for VMEAS, the unknown input voltage.

EQUATION 8: FINAL CONVERSION

In Equation 8, it is apparent that the measurement is
independent of the value of circuit elements R1 and C1.
This makes the conversion insensitive to errors in R1
and C1 value, due to the inaccuracy or temperature
variation. However, this does not mean that the value
of R1 and C1 is unimportant in the design of the A/D
converter. The values of R1 and C1 should be selected
based upon the number of bits of resolution. Looking
back at Equation 6 and solving R1C1 (see Equation 9):

Vref
R1

----------- C1
dV0
dt

----------=

Vmeas
R1

----------------- C1
dV0
dt

----------=

V0
1

R1C1
--------------- Vref td0

tref
=

V0
1

R1C1
--------------- Vmeas td0

tmeas
=

V0

Vref  tref
R1C1

-----------------------=

V0

Vmeas  tmeas
R1C1

-----------------------------------=

Vref tref
R1C1

-----------------------
Vmeas tmeas

R1C1
-----------------------------------=

Vmeas

Vref  tref
tmeas

-----------------------=

EQUATION 9: CALCULATION OF R1C1 VALUE

The actual value for R1C1 should be slightly smaller
than calculated, to ensure that the PIC16C5X or
PIC16F5X does not over count during the measure-
ment.

R1C1
Vmeas  tmeas

V0
-----------------------------------=

where

Vmeas Lowest  voltage  to  be measured at least  ten  LSBs =

tmeas Time  to  do  the  number of bits of  resolution  desired=

2Nx1/fosc x 4 clock/cycle x instruction cycles per  count.Where N is number of bit resolution 

V0 Vth  Threshold  Voltage  input  of the  PIC16C5x/PIC16F5x being  used  3V estimated =
DS00000513E-page 4  1997-2014 Microchip Technology Inc.

CIRCUIT CONFIGURATIONS
C and Assembly code implementing the circuit of
Figure 1 is listed in Appendix A: “Source Code in C”
and Appendix B: “Source Code in Assembly”.

These codes measure the time up to 16 bits. The full
implementation of algorithm can be seen in the
flowchart shown in Figure 5.

FIGURE 5: A/D CONVERSION FLOWCHART

There is a difference between the R1C1 value when
implementing in Assembly and C. This is because the
instruction cycles per count, when using C, is greater
than the Assembly.

Using the 4 MHz external RC clock oscillator, the value
of R1C1 can be calculated as follows (see
Equation 10):

For the C code:

EQUATION 10: R1C1 CALCULATION IN C

For the Assembly code (see Equation 11):

EQUATION 11: R1C1 CALCULATION IN ASSEMBLY

In actual applications, if measurement accuracy
permits, it may be advantageous to use lower
resolution bits and higher clock source. The math code
can be substantially reduced and the measure time is
reduced by the simpler code and shorter count.

CIRCUIT PERFORMANCE
The calibration value removes all first order errors
(offset, gain, R and C inaccuracy, power supply voltage
and temperature) except the reference voltage drift.
Any change in the reference voltage, including noise,
may result in measurement errors. Other error sources
may be analog switch leakage, resistor and capacitor
non-linearities, input threshold uncertainty and time
measurement uncertainty ( one instruction cycle
time). Measured performance shows the converter to
be accurate within 1% of full scale.

FIGURE 6: PERCENT ERROR AT FULL-SCALE CONVERSION

CONCLUSION
For a simple and low bandwidth analog application, it
usually requires a low cost yet high resolution A/D
converter. By using the PIC16C5X and PIC16F5X
baseline family of microcontrollers, this application
note demonstrates how to meet such requirements.
The A/D converter does not only use fewer
components but also has a capability to calibrate out
most circuit errors.

VIN
Computed Actual

%ERROR
HEX DEC HEX DEC

2.5 9A3 2467 9A1 2465 0.081%

VREF = 2.5V; R1 = 33K ohm; C1 = 1uF

APPENDIX A: SOURCE CODE IN C

Software License Agreement
The software supplied herewith by Microchip Technology Incorporated (the “Company”) is intended and supplied to you, the
Company’s customer, for use solely and exclusively with products manufactured by the Company.
The software is owned by the Company and/or its supplier, and is protected under applicable copyright laws. All rights are reserved.
Any use in violation of the foregoing restrictions may subject the user to criminal sanctions under applicable laws, as well as to civil
liability for the breach of the terms and conditions of this license.
THIS SOFTWARE IS PROVIDED IN AN “AS IS” CONDITION. NO WARRANTIES, WHETHER EXPRESS, IMPLIED OR STATU-
TORY, INCLUDING, BUT NOT LIMITED TO, IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICU-
LAR PURPOSE APPLY TO THIS SOFTWARE. THE COMPANY SHALL NOT, IN ANY CIRCUMSTANCES, BE LIABLE FOR
SPECIAL, INCIDENTAL OR CONSEQUENTIAL DAMAGES, FOR ANY REASON WHATSOEVER.
 1997-2014 Microchip Technology Inc. DS00000513E-page 7

EXAMPLE A-1: A/D CONVERTER PROGRAM IN C
/****************************************************************************

  File Name:

    PIC16F5xADC.c

  Summary:

    This is the main file used for the Application Note AN00513

    Hardware implementation requires:

    R1 = 33K ohm

    C1 = 1uF

    RC oscillator value are 4.7K ohm and 22pF, respectively.

    Vref = 2.5V

     

    Generation Information :

        Device            :  PIC16F57

        Compiler          :  XC8 v1.21

        MPLAB             :  MPLAB X 1.90

****************************************************************************/

/****************************************************************************

Copyright (c) 2013 released Microchip Technology Inc.  All rights reserved.

Microchip licenses to you the right to use, modify, copy and distribute

Software only when embedded on a Microchip microcontroller or digital signal

controller that is integrated into your product or third party product

(pursuant to the sublicense terms in the accompanying license agreement).

You should refer to the license agreement accompanying this Software for

additional information regarding your rights and obligations.

SOFTWARE AND DOCUMENTATION ARE PROVIDED "AS IS" WITHOUT WARRANTY OF ANY KIND,

EITHER EXPRESS OR IMPLIED, INCLUDING WITHOUT LIMITATION, ANY WARRANTY OF

MERCHANTABILITY, TITLE, NON-INFRINGEMENT AND FITNESS FOR A PARTICULAR PURPOSE.

IN NO EVENT SHALL MICROCHIP OR ITS LICENSORS BE LIABLE OR OBLIGATED UNDER

CONTRACT, NEGLIGENCE, STRICT LIABILITY, CONTRIBUTION, BREACH OF WARRANTY, OR

OTHER LEGAL EQUITABLE THEORY ANY DIRECT OR INDIRECT DAMAGES OR EXPENSES

INCLUDING BUT NOT LIMITED TO ANY INCIDENTAL, SPECIAL, INDIRECT, PUNITIVE OR

CONSEQUENTIAL DAMAGES, LOST PROFITS OR LOST DATA, COST OF PROCUREMENT OF

SUBSTITUTE GOODS, TECHNOLOGY, SERVICES, OR ANY CLAIMS BY THIRD PARTIES

(INCLUDING BUT NOT LIMITED TO ANY DEFENSE THEREOF), OR OTHER SIMILAR COSTS.

****************************************************************************/

#include <xc.h>

// CONFIG

#pragma config OSC = RC         // Oscillator selection bits (RC oscillator)

#pragma config WDT = OFF        // Watchdog timer enable bit (WDT disabled)

#pragma config CP = OFF         // Code protection bit (Code protection off)

void initPORTS()

{

    PORTA = 0x00;   //set RA0 low (on when activated)

    TRISB = 0x00;   //set all pins in PORTB as output

    OPTION = 0x28;  //select positive edge for TMR0
DS00000513E-page 8  1997-2014 Microchip Technology Inc.

EXAMPLE A-2: A/D CONVERTER PROGRAM IN C
}

void DSCHRG_CAP()

{

    TRISA = 0x00;   //RA0 on

    _delay(5000);   //RC discharging time

    TRISA = 0x01;   //set RA0 to have high impedance

}

void main()

{

    unsigned int Tcount, Tmeas, Tref, Vref, Vmeas;

    unsigned long  mpy;

    

    Vref = 0x09A3;      //Digital Equivalent of Vref voltage

    Vmeas = 0;          // Initiate Vmeas to 0

    initPORTS();        //Initialize I/O pin

    while(1)

    {

        PORTB = 0x05;   //S1 and S3 activated (Vref)

        DSCHRG_CAP();   //discharge capacitor charge

        PORTB = 0x09;   //S1 and S4 activated

        

        TMR0 = 0x00;    //reset counter and TMR0

        Tcount = 0;     

        

        while(!TMR0)    //increment Tcount until TMR0 trips

        {

            Tcount++;

        }

        Tref = Tcount;

        PORTB = 0x06;   //S2 and S3 activated (Vmeas)

        DSCHRG_CAP();   //discharge capacitor charge

        PORTB = 0x0A;   //S2 and S4 activated

        

        TMR0 = 0x00;    //reset counter and TMR0

        Tcount = 0;

        while(!TMR0)    //increment Tcount until TMR0 trips

        {

             Tcount++;  

        }

        Tmeas = Tcount;

        mpy = (long) Tref*Vref; //perform mathematical operation to get Vin

        Vmeas =  mpy / Tmeas;  //result

    }

}

 1997-2014 Microchip Technology Inc. DS00000513E-page 9

APPENDIX B: SOURCE CODE IN ASSEMBLY

EXAMPLE B-1: A/D CONVERTER PROGRAM IN ASSEMBLY
;****************************************************************************

;  Revision Date:

;    12-10-2013

;

;  File Name:

;    PIC16F5xADCasm.asm

;

;  Summary:

;    This is the main file used for the Application Note AN00513

;

;    Hardware implementation requires:

;    R1 = 13K ohm

;    C1 = 1uF

;    RC oscillator value are 4.7K ohm and 22pF, respectively.

;

;  Generation Information :

;        Device            :  PIC16F54

;        Compiler          :  MPASM v5.52

;        MPLAB             :  MPLAB X 1.90

;****************************************************************************

;****************************************************************************

;Copyright (c) 2013 released Microchip Technology Inc.  All rights reserved.

;

;Microchip licenses to you the right to use, modify, copy and distribute

;Software only when embedded on a Microchip microcontroller or digital signal

;controller that is integrated into your product or third party product

;(pursuant to the sublicense terms in the accompanying license agreement).

;

;You should refer to the license agreement accompanying this Software for

;additional information regarding your rights and obligations.

;

;SOFTWARE AND DOCUMENTATION ARE PROVIDED "AS IS" WITHOUT WARRANTY OF ANY KIND,

;EITHER EXPRESS OR IMPLIED, INCLUDING WITHOUT LIMITATION, ANY WARRANTY OF

;MERCHANTABILITY, TITLE, NON-INFRINGEMENT AND FITNESS FOR A PARTICULAR PURPOSE.

;IN NO EVENT SHALL MICROCHIP OR ITS LICENSORS BE LIABLE OR OBLIGATED UNDER

;CONTRACT, NEGLIGENCE, STRICT LIABILITY, CONTRIBUTION, BREACH OF WARRANTY, OR

;OTHER LEGAL EQUITABLE THEORY ANY DIRECT OR INDIRECT DAMAGES OR EXPENSES

;INCLUDING BUT NOT LIMITED TO ANY INCIDENTAL, SPECIAL, INDIRECT, PUNITIVE OR

;CONSEQUENTIAL DAMAGES, LOST PROFITS OR LOST DATA, COST OF PROCUREMENT OF

;SUBSTITUTE GOODS, TECHNOLOGY, SERVICES, OR ANY CLAIMS BY THIRD PARTIES

;(INCLUDING BUT NOT LIMITED TO ANY DEFENSE THEREOF), OR OTHER SIMILAR COSTS.

;****************************************************************************

#INCLUDE <P16F54.INC>

 __config 0xFFFB                ;configuration setting: 

                                ; OSC_RC, WDT_OFF, CP_OFF

                                ;Digital Equivalent of Vref voltage

#DEFINE VCALMS        009       ;VCAL MSB VALUE IN HEX

#DEFINE VCALLS        02E       ;VCAL LSB VALUE IN HEX
DS00000513E-page 10  1997-2014 Microchip Technology Inc.

EXAMPLE B-2: A/D CONVERTER PROGRAM IN ASSEMBLY
ACCA     EQU     8

ACCB     EQU     0A

ACCC     EQU     0C

ACCD     EQU     0E

ACCE     EQU     10

TMEAS    EQU     12

TEMP     EQU     14

TEMP1    EQU     16

TEMP2    EQU     17

TEMP3    EQU     18

         ORG     0

         GOTO    VOLTS           ;PROGRAM CODE

MADD     MOVF    ACCA+1,W

         ADDWF   ACCB+1, F       ;ADD LSB

         BTFSC   3,0             ;ADD IN CARRY

         INCF    ACCB, F

         MOVF    ACCA,W

         ADDWF   ACCB, F         ;ADD MSB

         RETLW   0

         NOP

MPY      CALL    SETUP           ;RESULTS IN B(16 MSB'S) AND C(16 LSB'S)

MLOOP    RRF     ACCD, F         ;ROTATE D RIGHT

         RRF     ACCD+1, F

         SKPNC                   ;NEED TO ADD?

         CALL    MADD

         RRF     ACCB, F

         RRF     ACCB+1, F

         RRF     ACCC, F

         RRF     ACCC+1, F

         DECFSZ  TEMP, F         ;LOOP UNTIL ALL BITS CHECKED

         GOTO    MLOOP

         RETLW   0

         NOP

SETUP    MOVLW   10

         MOVWF   TEMP

         MOVF    ACCB,W          ;MOVE B TO D

         MOVWF   ACCD

         MOVF    ACCB+1,W

         MOVWF   ACCD+1

         MOVF    ACCC,W

         MOVWF   ACCE

         MOVF    ACCC+1,W

         MOVWF   ACCE+1

 CLRF    ACCB

         CLRF    ACCB+1

         RETLW   0
 1997-2014 Microchip Technology Inc. DS00000513E-page 11

EXAMPLE B-3: A/D CONVERTER PROGRAM IN ASSEMBLY
         NOP

DIV      CALL    SETUP

         MOVLW   20

         MOVWF   TEMP

         CLRF    ACCC

         CLRF    ACCC+1

DLOOP    CLRC

         RLF     ACCE+1, F

         RLF     ACCE, F

         RLF     ACCD+1, F

         RLF     ACCD, F

         RLF     ACCC+1, F

         RLF     ACCC, F

         MOVF    ACCA,W

         SUBWF   ACCC,W          ;CHECK IF A>C

         SKPZ

         GOTO    NOCHK

         MOVF    ACCA+1,W

         SUBWF   ACCC+1,W        ;IF MSB EQUAL THEN CHECK LSB

NOCHK    SKPC                    ;CARRY SET IF C>A

         GOTO    NOGO

         MOVF    ACCA+1,W        ;C-A INTO C

         SUBWF   ACCC+1, F

         BTFSS   3,0

         DECF    ACCC, F

         MOVF    ACCA,W

         SUBWF   ACCC, F

         SETC                    ;SHIFT A 1 INTO B (RESULT)

NOGO     RLF     ACCB+1, F

         RLF     ACCB, F

         DECFSZ  TEMP, F         ;LOOP UNTILL ALL BITS CHECKED

         GOTO    DLOOP

         RETLW   0

DSCHRG   MOVLW   B'00001110'     ;DISCHARGE C (RA0 ON)

         TRIS    5

         MOVLW   0FF

         MOVWF   TEMP1

         MOVWF   TEMP2

         MOVWF   TEMP3

LOOP1    DECFSZ  TEMP1, F         ;WAIT

         GOTO    LOOP1

LOOP2    DECFSZ  TEMP2, F         ;WAIT

         GOTO    LOOP2

LOOP3    DECFSZ  TEMP3, F         ;WAIT

         GOTO    LOOP3

         MOVLW   B'00001111'     ;ALL RA HIGH Z

         TRIS    5

         RETLW   0

M_TIME   CLRF    1               ;CLEAR RTCC REGISTER

         CLRF    ACCA+1          ;CLEAR 16 BIT COUNTER

         CLRF    ACCA
DS00000513E-page 12  1997-2014 Microchip Technology Inc.

EXAMPLE B-4: A/D CONVERTER PROGRAM IN ASSEMBLY
TLOOP    INCFSZ  ACCA+1, F

         GOTO    ENDCHK

         INCFSZ  ACCA, F

         GOTO    ENDCHK

         GOTO    END_M

ENDCHK   BTFSS   1,0             ;CHECK FOR RTCC TRIP

         GOTO    TLOOP

END_M    MOVF    1,W

         RETLW   0

VOLTS    MOVLW   B'00000110'     ;SET S2 AND S3 HIGH(ON WHEN ACTIVATED)

         MOVWF   6

         MOVLW   B'00000000'     ;ACTIVATE SWITCHES S1-S4

         TRIS    6

         MOVLW   B'00101000'     ;SELECT POSITIVE EDGE FOR RTCC

         OPTION

         MOVLW   B'00000000'

         MOVWF   5               ;SET RA0 LOW (ON WHEN ACTIVATED)

MEAS     CALL    DSCHRG          ;CHARGE CAPACITOR TO VIN

         MOVLW   B'00001010'     ;S2 AND S4 ON

         MOVWF   6

         CALL    M_TIME          ;MEASURE TIME

         MOVF    ACCA+1,W

         MOVWF   TMEAS+1         ;STORE LSB

         MOVF    ACCA,W

         MOVWF   TMEAS           ;STORE MSB

CAL      MOVLW   B'00000101'     ;S1 AND S3 ON

         MOVWF   6

         CALL    DSCHRG          ;CHARGE CAPACITOR TO VREF

         MOVLW   B'00001001'     ;S1 AND S4 ON

         MOVWF   6

         CALL    M_TIME          ;MEASURE TIME

         MOVLW   VCALLS

         MOVWF   ACCB+1

         MOVLW   VCALMS

         MOVWF   ACCB

         CALL    MPY             ;MULTIPLY ACCA(TCAL) * ACCB(VREF)

         MOVF    TMEAS+1,W

         MOVWF   ACCA+1

         MOVF    TMEAS,W

         MOVWF   ACCA

         CALL    DIV             ;DIVIDE ACCB(TCAL * V) BY ACCA(TMEAS)

         GOTO   VOLTS

         END
 1997-2014 Microchip Technology Inc. DS00000513E-page 13

NOTES:

Note the following details of the code protection feature on Microchip devices:
• Microchip products meet the specification contained in their particular Microchip Data Sheet.

• Microchip believes that its family of products is one of the most secure families of its kind on the market today, when used in the 
intended manner and under normal conditions.

• There are dishonest and possibly illegal methods used to breach the code protection feature. All of these methods, to our 
knowledge, require using the Microchip products in a manner outside the operating specifications contained in Microchip’s Data 
Sheets. Most likely, the person doing so is engaged in theft of intellectual property.

• Microchip is willing to work with the customer who is concerned about the integrity of their code.

• Neither Microchip nor any other semiconductor manufacturer can guarantee the security of their code. Code protection does not 
mean that we are guaranteeing the product as “unbreakable.”

Code protection is constantly evolving. We at Microchip are committed to continuously improving the code protection features of our
products. Attempts to break Microchip’s code protection feature may be a violation of the Digital Millennium Copyright Act. If such acts
allow unauthorized access to your software or other copyrighted work, you may have a right to sue for relief under that Act.
Information contained in this publication regarding device
applications and the like is provided only for your convenience
and may be superseded by updates. It is your responsibility to
ensure that your application meets with your specifications.
MICROCHIP MAKES NO REPRESENTATIONS OR
WARRANTIES OF ANY KIND WHETHER EXPRESS OR
IMPLIED, WRITTEN OR ORAL, STATUTORY OR
OTHERWISE, RELATED TO THE INFORMATION,
INCLUDING BUT NOT LIMITED TO ITS CONDITION,
QUALITY, PERFORMANCE, MERCHANTABILITY OR
FITNESS FOR PURPOSE. Microchip disclaims all liability
arising from this information and its use. Use of Microchip
devices in life support and/or safety applications is entirely at
the buyer’s risk, and the buyer agrees to defend, indemnify and
hold harmless Microchip from any and all damages, claims,
suits, or expenses resulting from such use. No licenses are
conveyed, implicitly or otherwise, under any Microchip
intellectual property rights.

